MODULE *Paxos*

This is a specification of the *Paxos* algorithm without explicit leaders or learners. It refines the spec in *Voting*.

In this version:

− $Phase1a(b)$ messages can be duplicated (realized as adding it to a set in TLA+).

− $Phase1b(a)$: Due to the enabling condition "$\land\ m.bal > maxBal(a)$", the "1b" messages are sent conditionally.

− $Phase2a(b,\ v)$ is enabled only if the proposer of b has not issued "2a" messages before. That is, ballots are "client-restricted" in terms of $GS2Paxos@arXiv2019$. Thus, $OneValuePerBallot$ (and hence, $OneVote$) holds.

− $Phase2b(a)$: Due to the enabling condition "$\land\ m.bal \geq maxBal[a]$", the acceptors can accept a ballot only if it has not promised not to do that.

EXTENDS *Integers*, *TLC*

CONSTANT *Value*, *Acceptor*, *Quorum*

ASSUME $QuorumAssumption \triangleq$
$\quad \land \quad \forall\, Q \in Quorum : Q \subseteq Acceptor$
$\quad \land \quad \forall\, Q1,\ Q2 \in Quorum : Q1 \cap Q2 \neq \{\}$

$Ballot \triangleq Nat$
$None \triangleq$ CHOOSE $v : v \notin Ballot$

$Message \triangleq$
$\qquad [type : \{\text{"1a"}\},\ bal : Ballot]$
$\quad \cup \quad [type : \{\text{"1b"}\},\ acc : Acceptor,\ bal : Ballot,$
$\qquad\ mbal : Ballot \cup \{-1\},\ mval : Value \cup \{None\}]$
$\quad \cup \quad [type : \{\text{"2a"}\},\ bal : Ballot,\ val : Value]$
$\quad \cup \quad [type : \{\text{"2b"}\},\ acc : Acceptor,\ bal : Ballot,\ val : Value]$

VARIABLE $maxBal$, $maxVBal$, $maxVal$, $msgs$

$maxBal[a]$: the largest ballot number a has seen
$\langle maxVBal[a],\ maxVal[a]\rangle$ is the vote with the largest ballot number cast by a; it is $\langle -1,\ None\rangle$ if a has not cast any vote.
The set of all messages that have been sent.

$Send(m) \triangleq msgs' = msgs \cup \{m\}$

$vars \triangleq \langle maxBal,\ maxVBal,\ maxVal,\ msgs\rangle$

NOTE: The algorithm is easier to understand in terms of the set *msgs* of all messages that have ever been sent. A more accurate model would use one or more variables to represent the messages actually in transit, and it would include actions representing message loss and duplication as well as message receipt.

In the current spec, there is no need to model message loss because we are mainly concerned with the algorithm's safety property. The safety part of the spec says only what messages may be received and does not assert that any message actually is received. Thus, there is no difference between a lost message and one that is never received. The liveness property of the spec that we check makes it clear what messages must be received (and hence either not lost or successfully retransmitted if lost) to guarantee progress.

```
63  TypeOK ≜
64      ∧  maxBal ∈ [Acceptor → Ballot ∪ {−1}]
65      ∧  maxVBal ∈ [Acceptor → Ballot ∪ {−1}]
66      ∧  maxVal ∈ [Acceptor → Value ∪ {None}]
67      ∧  msgs ⊆ Message
68  ----------------------------------------------------------------
69  Init ≜
70      ∧ maxBal = [a ∈ Acceptor ↦ −1]
71      ∧ maxVBal = [a ∈ Acceptor ↦ −1]
72      ∧ maxVal = [a ∈ Acceptor ↦ None]
73      ∧ msgs = {}
74  ----------------------------------------------------------------
```

In an implementation, there will be a leader process that orchestrates a ballot. The ballot b leader performs actions $Phase1a(b)$ and $Phase2a(b)$. The $Phase1a(b)$ action sends a phase $1a$ message that begins ballot b.

```
80  Phase1a(b) ≜
81      ∧   Send([type ↦ "1a", bal ↦ b])
82      ∧   UNCHANGED ⟨maxBal, maxVBal, maxVal⟩
```

Upon receipt of a ballot b phase $1a$ message, acceptor a can perform a $Phase1b(a)$ action only if $b > maxBal[a]$. The action sets $maxBal[a]$ to b and sends a phase $1b$ message to the leader containing the values of $maxVBal[a]$ and $maxVal[a]$.

```
88  Phase1b(a) ≜
89      ∧   ∃ m ∈ msgs :
90            ∧ m.type = "1a"
91            ∧ m.bal > maxBal[a]
92            ∧ maxBal' = [maxBal EXCEPT ![a] = m.bal]      make promise
93            ∧ Send([type ↦ "1b", acc ↦ a, bal ↦ m.bal,
94                    mbal ↦ maxVBal[a], mval ↦ maxVal[a]])
95      ∧   UNCHANGED ⟨maxVBal, maxVal⟩

97  NoBackInTime ≜
98      ∀ m ∈ msgs : m.type = "1b" ⇒ m.mbal < m.bal
```

The $Phase2a(b, v)$ action can be performed by the ballot b leader if two conditions are satisfied: (i) it has not already performed a phase $2a$ action for ballot b and (ii) it has received ballot b phase $1b$ messages from some quorum Q from which it can deduce that the value v is safe at ballot b. These enabling conditions are the first two conjuncts in the definition of $Phase2a(b, v)$. The second conjunct, expressing condition (ii), is the heart of the algorithm. To understand it, observe that the existence of a phase $1b$ message m in $msgs$ implies that $m.mbal$ is the highest ballot number less than $m.bal$ in which acceptor $m.acc$ has or ever will cast a vote, and that $m.mval$ is the value it voted for in that ballot if $m.mbal \neq -1$. It is not hard to deduce from this that the second conjunct implies that there exists a quorum Q such that $ShowsSafeAt(Q, b, v)$ (where $ShowsSafeAt$ is defined in module $Voting$).

The action sends a phase $2a$ message that tells any acceptor a that it can vote for v in ballot b, unless it has already set $maxBal[a]$ greater than b (thereby promising not to vote in ballot b).

```
P2C(b, v) ≜
    ∃ Q ∈ Quorum :
      LET Q2bv ≜ {m ∈ msgs : m.type = "2b" ∧ m.acc ∈ Q ∧ m.bal < b}
      IN  ∨ Q2bv = {}
          ∨ ∃ m ∈ Q2bv :
              ∧ m.val = v
              ∧ ∀ mm ∈ Q2bv : m.bal ≥ mm.bal

Phase2a(b, v) ≜
  ∧ ¬∃ m ∈ msgs : m.type = "2a" ∧ m.bal = b
  ∧ ∃ Q ∈ Quorum :
      LET Q1b ≜ {m ∈ msgs : m.type = "1b" ∧ m.acc ∈ Q ∧ m.bal = b}
          Q1bv ≜ {m ∈ Q1b : m.mbal ≥ 0}
      IN  ∧ ∀ a ∈ Q : ∃ m ∈ Q1b : m.acc = a
          ∧ ∨ Q1bv = {}
            ∨ ∃ m ∈ Q1bv :
                ∧ m.mval = v
                ∧ ∀ mm ∈ Q1bv : m.mbal ≥ mm.mbal
  ∧ Send([type ↦ "2a", bal ↦ b, val ↦ v])
  ∧ Assert(P2C(b, v), "P2C Fails!")
  ∧ UNCHANGED ⟨maxBal, maxVBal, maxVal⟩
```

The $Phase2b(a)$ action is performed by acceptor a upon receipt of a phase $2a$ message. Acceptor a can perform this action only if the message is for a ballot number greater than or equal to $maxBal[a]$. In that case, the acceptor votes as directed by the phase $2a$ message, setting $maxBVal[a]$ and $maxVal[a]$ to record that vote and sending a phase $2b$ message announcing its vote.

Note: It also sets $maxBal[a]$ to the message's ballot number. Otherwise,
(1) *NoBackInTime* for $Phase1b$ does not hold.
(2) "Non-Increasing Error" assertion in $Phase2b(a)$ fails.
(3) $P2C$ assertion for $Phase2a$ does not hold ???

```
Phase2b(a) ≜
    ∃ m ∈ msgs :
      ∧ m.type = "2a"
      ∧ m.bal ≥ maxBal[a]
      ∧ maxBal' = [maxBal EXCEPT ![a] = m.bal]
      ∧ maxVBal' = [maxVBal EXCEPT ![a] = m.bal]
      ∧ Assert(maxVBal'[a] ≥ maxVBal[a], "Non-Increasing Error!")
      ∧ maxVal' = [maxVal EXCEPT ![a] = m.val]
      ∧ Send([type ↦ "2b", acc ↦ a, bal ↦ m.bal, val ↦ m.val])
```

In an implementation, there will be learner processes that learn from the phase $2b$ messages if a value has been chosen. The learners are omitted from this abstract specification of the algorithm.

```
Next ≜
    ∨ ∃ b ∈ Ballot :
```

$$\lor\ Phase1a(b)$$
$$\lor\ \exists\, v \in Value : Phase2a(b,\, v)$$
$$\lor\ \exists\, a \in Acceptor : Phase1b(a) \lor Phase2b(a)$$

$$Spec \triangleq Init \land \Box[Next]_{vars}$$

We now define the refinement mapping under which this algorithm implements the specification in module *Voting*.

As we observed, votes are registered by sending phase $2b$ messages. So the array *votes* describing the votes cast by the acceptors is defined as follows.

$$votes \triangleq [a \in Acceptor \mapsto \{\langle m.bal,\, m.val\rangle : m \in \{mm \in msgs : \land\, mm.type = \text{“2b”} \land\, mm.acc = a\}\}]$$

We now instantiate module *Voting*, substituting the constants *Value*, *Acceptor*, and *Quorum* declared in this module for the corresponding constants of that module *Voting*, and substituting the variable *maxBal* and the defined state function *votes* for the correspondingly-named variables of module *Voting*.

$$V \triangleq \text{INSTANCE } Voting$$

$Consistency \triangleq V!C!Inv$ Only about “chosen”: $TypeOK \land Cardinality(chosen) \leq 1$

$StrongConsistency \triangleq V!Inv$ $TypeOK \land VotesSafe \land OneValuePerBallot$

$$\text{THEOREM } Spec \Rightarrow V!Spec$$

Here is a first attempt at an inductive invariant used to prove this theorem.

$$\begin{aligned}
Inv \triangleq\ & \land\, TypeOK \\
& \land\, \forall\, a \in Acceptor : \text{IF } maxVBal[a] = -1 \\
& \qquad\qquad \text{THEN } maxVal[a] = None \\
& \qquad\qquad \text{ELSE } \langle maxVBal[a],\, maxVal[a]\rangle \in votes[a] \\
& \land\, \forall\, m \in msgs : \\
& \quad \land\, (m.type = \text{“1b”}) \Rightarrow\ \land\, maxBal[m.acc] \geq m.bal \\
& \qquad\qquad\qquad\qquad\ \land\, (m.mbal \geq 0) \Rightarrow \\
& \qquad\qquad\qquad\qquad\quad \langle m.mbal,\, m.mval\rangle \in votes[m.acc] \\
& \quad \land\, (m.type = \text{“2a”}) \Rightarrow\ \land\, \exists\, Q \in Quorum : \\
& \qquad\qquad\qquad\qquad\qquad\ V!ShowsSafeAt(Q,\, m.bal,\, m.val) \\
& \qquad\qquad\qquad\qquad\ \land\, \forall\, mm \in msgs : \land\, mm.type = \text{“2a”} \\
& \qquad\qquad\qquad\qquad\qquad\qquad\ \land\, mm.bal\ = m.bal \\
& \qquad\qquad\qquad\qquad\qquad\qquad\ \Rightarrow mm.val = m.val \\
& \land\, V!Inv
\end{aligned}$$
